



Application Note #2445

Stepper Position Maintenance Mode

Stepper motors are designed to offer an economical method of open-loop motion control. While easy-to-use, they are prone to skipping or stalling resulting in position loss. This application note discusses various options for handling stepper motor position error and presents new commands available in Galil controllers to implement these options.

In an attempt to "close-the-loop" and obtain actual position information, a feedback sensor is often added to a stepper motor. This position information is used to determine if there is any significant difference between the commanded and the actual motor positions. If such error is detected, the designer has several options for dealing with the error including halting the motion, executing a correction move, or continuing motion with error.

Galil has added several commands to its controllers for the new Stepper Position Maintenance Mode (SPM), allowing the user to actively handle stepper position error.

New Commands

SPM mode is configured, executed, and managed with six new commands and one existing command. This mode also utilizes the #POSERR automatic subroutine allowing for automatic user-defined handling of an error event. These new commands are as follows:

Internal Controller Commands (user can query):

QS Error Magnitude (pulses)

User Configurable Commands (user can query & change):

OE Profiler Off-On Error

YA Step Drive Resolution (pulses / full motor step)

YB Step Motor Resolution (full motor steps / revolution)

YC Encoder Resolution (counts / revolution)

YR Error Correction (pulses)

YS Stepper Position Maintenance enable, status

A step pulse is defined by the resolution of the step drive being used. Therefore, one pulse could be a full step, a half step or a microstep.

When a Galil controller is configured for step motor operation, the step pulses output by the controller are internally fed back into the auxiliary encoder register. The feedback encoder on the stepper is connected to the main encoder port. Enabling the SPM mode on a controller with YS=1 executes an internal monitoring of the auxiliary and main encoder registers for that axis or axes. Position error is then tracked in step pulses between these two registers (QS command).

$$QS = TD - \frac{TP \times YA \times YB}{YC}$$

Where TD is the auxiliary encoder register(step pulses) and TP is the main encoder register(feedback encoder). Additionally, YA defines the step drive resolution where YA=1 for full stepping, YA=2 for half stepping or YA=64 for microstepping with the SDM-20640 microstepping drive.

Error Limit

Typically, if a stepper falls behind by more than two full steps, the error is unrecoverable. Thus, the value of QS is internally monitored to determine if it exceeds a preset limit of three full motor steps. Once the value of QS exceeds this limit, the controller then performs the following actions:

1. The motion is maintained or is stopped, depending on the setting of the OE command. If OE=0 the axis stays in motion, if OE=1 the axis is stopped.
2. YS is set to 2, which causes the automatic subroutine labeled #POSERR to be executed.

Correction

A correction move can be easily commanded by assigning the value of QS to the YR correction move command. The correction move is issued only after the axis has been stopped. After an error correction move has completed and QS is less than three full motor steps, the YS error status bit is automatically reset back to 1 indicating a cleared error.

Examples

Consider the DMC-2143 motion controller with 1.8° step motors, and 4000 count/rev encoders. The following code demonstrates what is necessary to set up SPM mode for a full step drive, a half step drive, and then the SDM-20640 1/64th step microstepping drive. The necessary difference is with the YA command. (comments are spaced away from commands for clarity)

Full-Stepping Drive, X axis:

```
#SETUP
OE1;           'SET THE PROFILER TO STOP AXIS UPON ERROR
KS16;         'SET STEP SMOOTHING
```

```

MT-2;          'MOTOR TYPE SET TO STEPPER
YA1;           'STEP RESOLUTION OF THE SDM-20240 (FULLSTEP)
YB200;        'MOTOR RESOLUTION (FULL STEPS PER REVOLUTION)
YC4000;       'ENCODER RESOLUTION (COUNTS PER REVOLUTION)
SHX;          'ENABLE AXIS
WT50;         'ALLOW SLIGHT SETTLE TIME
YS1;          'ENABLE SPM MODE

```

Half-Stepping Drive, X axis:

```

#SETUP
OE1;          'SET THE PROFILER TO STOP AXIS UPON ERROR
KS16;         'SET STEP SMOOTHING
MT-2;         'MOTOR TYPE SET TO STEPPER
YA2;          'STEP RESOLUTION OF THE SDM-20240 (HALF-STEP)
YB200;        'MOTOR RESOLUTION (FULL STEPS PER REVOLUTION)
YC4000;       'ENCODER RESOLUTION (COUNTS PER REVOLUTION)
SHX;          'ENABLE AXIS
WT50;         'ALLOW SLIGHT SETTLE TIME
YS1;          'ENABLE SPM MODE

```

1/64th Step Microstepping Drive, X axis:

```

#SETUP
OE1;          'SET THE PROFILER TO STOP AXIS UPON ERROR
KS16;         'SET STEP SMOOTHING
MT-2;         'MOTOR TYPE SET TO STEPPER
YA64;         'STEP RESOLUTION OF THE SDM-20640 (MICROSTEP)
YB200;        'MOTOR RESOLUTION (FULL STEPS PER REVOLUTION)
YC4000;       'ENCODER RESOLUTION (COUNTS PER REVOLUTION)
SHX;          'ENABLE AXIS
WT50;         'ALLOW SLIGHT SETTLE TIME
YS1;          'ENABLE SPM MODE

```

Now consider the DMC-2143 motion controller with an SDM-20240 step drive setup for half stepping, and the same motors and encoders. The following code demonstrates what is necessary to set up SPM mode for the X axis, detect error, stop the motor, correct the error, and return to the main code.

```

#SETUP
OE1;          'SET THE PROFILER TO STOP AXIS UPON ERROR
KS16;         'SET STEP SMOOTHING
MT-2,-2,-2,-2; 'MOTOR TYPE SET TO STEPPER
YA2;          'STEP RESOLUTION OF THE SDM-20240 (HALF-STEP)
YB200;        'MOTOR RESOLUTION (FULL STEPS PER REVOLUTION)
YC4000;       'ENCODER RESOLUTION (COUNTS PER REVOLUTION)
SHX;          'ENABLE AXIS
WT100;        'ALLOW SLIGHT SETTLE TIME
YS1;          'ENABLE SPM MODE

#MOTION
SP512;        'SET THE SPEED
PR1000;       'PREPARE MODE OF MOTION
BGX;          'BEGIN MOTION
#LOOP;JP#LOOP; 'KEEP THREAD ZERO ALIVE FOR #POSERR TO RUN IN

```

REM When error occurs, the axis will stop due to OE1. In
 REM #POSERR, query the status YS and error QS, correct,
 REM and return to the main code.

```
#POSERR;          'AUTOMATIC SUBROUTINE IS CALLED WHEN YS=2
WT100;           'WAIT HELPS USER SEE THE CORRECTION
spsave=_SPX;     'SAVE CURRENT SPEED SETTING
JP#RETURN,_YSX<>2;'RETURN TO THREAD ZERO IF INVALID ERROR
SP64;           'SET SLOW SPEED SETTING FOR CORRECTION
MG"ERROR= ",_QX
YRX=_QX;         'ELSE, ERROR IS VALID, USE QS FOR CORRECTION
MCX;            'WAIT FOR MOTION TO COMPLETE
MG"CORRECTED, ERROR NOW= ",_QX
WT100;          'WAIT HELPS USER SEE THE CORRECTION

#RETURN
SPX=spsave;      'RETURN THE SPEED TO PREVIOUS SETTING
RE0;            'RETURN FROM #POSERR
```

In addition to correcting for error due to stalling or skipping, the SPM mode can be used as a method to correct for friction at the end of a microstepping move. This capability provides closed-loop control at the application program level. For example, consider the DMC-2143 controller and SDM-20640 microstepping drive with 1.8° step motors, and 4000 count per revolution encoders. The following example illustrates how the SPM commands can be useful in correcting for X axis friction after each move during reciprocating motion.

```
#SETUP;          'SET THE PROFILER TO CONTINUE UPON ERROR
KS16;           'SET STEP SMOOTHING
MT-2,-2,-2,-2;  'MOTOR TYPE SET TO STEPPER
YA64;          'STEP RESOLUTION OF THE SDM-20640 (MICROSTEP)
YB200;         'MOTOR RESOLUTION (FULL STEPS PER 'REVOLUTION)
YC4000;        'ENCODER RESOLUTION (COUNTS PER REVOLUTION)
SHX;           'ENABLE AXIS
WT50;          'ALLOW SLIGHT SETTLE TIME
YS1;           'ENABLE SPM MODE

#MOTION;         'PERFORM MOTION
SP16384;        'SET THE SPEED
PR10000;        'PREPARE MODE OF MOTION
BGX;           'BEGIN MOTION
MCX
JS#CORRECT;     'MOVE TO CORRECTION
#MOTION2
SP16384;        'SET THE SPEED
PR-10000;       'PREPARE MODE OF MOTION
BGX;           'BEGIN MOTION
MCX
JS#CORRECT;     'MOVE TO CORRECTION
JP#MOTION

#CORRECT;       'CORRECTION CODE
spx=_SPX
#LOOP;         'SAVE SPEED VALUE
```

```

SP2048;          'SET A NEW SLOW CORRECTION SPEED
WT100;           'STABILIZE
JP#END,@ABS[_QX]<10;'END CORRECTION IF ERROR IS WITHIN DEFINED
                  'TOLERANCE
YRX=_QX;         'CORRECTION MOVE
MCX
JP#LOOP;         'KEEP CORRECTING UNTIL ERROR IS WITHIN
                  'TOLERANCE
#END;            'END #CORRECT, RETURN TO CODE
SPX=spX
EN

```

For more information on Stepper Position Maintenance please contact a Galil Applications Engineer. For more information on step motor performance, please see the following resources.

Step Motor Performance Limitations:

<http://www.galilmc.com/support/appnotes/miscellaneous/note5466.pdf>

Microstepping Performance:

<http://www.galilmc.com/support/appnotes/miscellaneous/note5452.pdf>

Web Tutorials on Full & Microstepping:

<http://galilmc.com/training/webconf.html>